/*
 * Copyright (c) 2020, WSO2 Inc. (http://www.wso2.org) All Rights Reserved.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 * http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 *
 */

import org.apache.tools.ant.taskdefs.condition.Os

task updateVersionTesterinaReportTools(type: Exec) {
    doFirst {
        println "update version to " + project.version
        if (Os.isFamily(Os.FAMILY_WINDOWS)) {
            commandLine 'cmd', '/c', 'npm', 'run', 'update-version', '--', project.version
        } else {
            commandLine 'npm', 'run', 'update-version', '--', project.version
        }
    }
}

task npmInstallTesterinaReportTools(type: Exec) {
    dependsOn updateVersionTesterinaReportTools
    if (Os.isFamily(Os.FAMILY_WINDOWS)) {
        commandLine 'cmd', '/c', 'npm', 'install'
    } else {
        commandLine 'npm', 'install'
    }
}

task npmBuildTesterinaReportTools(type: Exec) {
    dependsOn npmInstallTesterinaReportTools
    if (Os.isFamily(Os.FAMILY_WINDOWS)) {
        commandLine 'cmd', '/c', 'npm', 'run', 'build'
    } else {
        commandLine 'npm', 'run', 'build'
    }
}

task createTesterinaReportToolsZip(type: Zip) {
    dependsOn npmBuildTesterinaReportTools
    from "build"
    include 'index.html'
    include 'favicon.ico'
    include 'manifest.json'
    include '**/static/css/*.css'
    include '**/static/js/*.js'
    exclude '**/static/js/runtime-*.js'
    archiveFileName = 'testerina-report-tools-' + project.version + '.zip'
    destinationDirectory = file("$project.rootDir/build/tests/")
}
